// File: Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(
        address indexed previousOwner,
        address indexed newOwner
    );

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(
            newOwner != address(0),
            "Ownable: new owner is the zero address"
        );
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}

// File: RewardBox.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.9;

import "./Ownable.sol";

interface IRewardToken {
    function mint(address account, uint256 amount) external;
}

contract RewardBox is Ownable {
    string public constant CONTRACT_NAME = "RewardBox";
    bytes32 public constant DOMAIN_TYPEHASH =
        keccak256(
            "EIP712Domain(string name,uint256 chainId,address verifyingContract)"
        );
    bytes32 public constant WITHDRAW_TYPEHASH =
        keccak256(
            "Withdraw(uint256[] withdrawalIds,uint256[] amounts,address user)"
        );

    IRewardToken public rewardToken;

    address public admin = 0x019dCf8781F46Cb0880E3Ee68E1Aad165827FA04;

    mapping(uint256 => bool) public withdrawal;

    event Withdraw(uint256 withdrawalId, uint256 amount, address user);

    constructor() {}

    function setRewardToken(IRewardToken _rewardToken) external onlyOwner {
        rewardToken = _rewardToken;
    }

    function changeAdmin(address newAdmin) external onlyOwner {
        admin = newAdmin;
    }

    function withdraw(
        uint256[] calldata withdrawalIds,
        uint256[] calldata amounts,
        address user,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external {
        bytes32 domainSeparator = keccak256(
            abi.encode(
                DOMAIN_TYPEHASH,
                keccak256(bytes(CONTRACT_NAME)),
                getChainId(),
                address(this)
            )
        );
        bytes32 structHash = keccak256(
            abi.encode(
                WITHDRAW_TYPEHASH,
                keccak256(abi.encodePacked(withdrawalIds)),
                keccak256(abi.encodePacked(amounts)),
                user
            )
        );
        bytes32 digest = keccak256(
            abi.encodePacked("\x19\x01", domainSeparator, structHash)
        );
        address signatory = ecrecover(digest, v, r, s);
        require(signatory == admin, "Invalid signatory");

        uint256 total = 0;
        for (uint256 i = 0; i < withdrawalIds.length; i++) {
            uint256 id = withdrawalIds[i];
            if (!withdrawal[id] && amounts[i] > 0) {
                total += amounts[i];
                withdrawal[id] = true;
                emit Withdraw(id, amounts[i], user);
            }
        }

        if (total > 0) {
            rewardToken.mint(user, total);
        }
    }

    function getChainId() internal view returns (uint256) {
        uint256 chainId;
        assembly {
            chainId := chainid()
        }
        return chainId;
    }
}